



Techniques d'intégration (partie 2 de 5)

Par parties

© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

Ce document est une révision de celui produit en 2003. L'objectif principal de ce document est de rendre l'élève apte à transposer en Maple la technique d'intégration par parties.

$$\int u \, dv = uv - \int v \, du$$

Cette transposition sera réalisée avec certaines macro-commandes de la sous-bibliothèque [Calculus1](#) de la bibliothèque [Student](#).


Les exemples développés dans ce document illustreront la manière d'obtenir l'intégrale indéfinie de deux façons:

- une première façon consistera à obtenir directement l'intégrale indéfinie avec la macro-commande [int](#) de la bibliothèque principale, question d'observer le mécanisme de la simplification automatique de Maple.
- la seconde façon, quant à elle, consistera à faire un développement pas à pas tel un développement *manus scriptus*. Un tel développement sera réalisé selon les règles d'intégration [Rule](#) de *Student*[*Calculus1*].

Pour compléter tout développement pas à pas, la dernière étape consistera à faire une vérification de la réponse finale par dérivation.

Attention:

- S'assurer que la rubrique « Afficher les étiquettes des équations » soit cochée: voir *menu Outils* → *Options...* → *Affichage*.
- S'assurer que le préfixe de numérotation d'étiquettes soit initialisée à 1 avec un schéma de numérotation séquentielle: voir *menu Format* → *Étiquettes d'équation (L)* → *Affichage des étiquettes...* (L)



- L'insertion d'une étiquette est obtenu avec le raccourci « ctrl + 1 » (1 pour *label* en anglais).

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;  
with(Student[Calculus1],Rule,Integrand);
```

```
infolevel[Student[Calculus1]]:=1:
                                     [Rule,Integrand] (1)
```

Assurance que les étiquettes des équations soient affichées.

```
> interface(labelling=true)
                                     true (2)
```

Réglage de l'affichage pour les variables conditionnées.

```
> with(Typesetting):
   interface(typesetting=standard); #niveau de composition Maple
   Standard
   interface(showassumed=2);      # Variables avec suppositiosnOhrase
                                     extended, [extended]
                                     2 (3)
```

Intégration par parties

La technique d'intégration par parties repose sur le développement de la différentielle d'un produit. Soit u et v , deux fonctions dérivables. Puisque $d(uv) = duv + u dv$, nous avons

$$\int d(uv) = \int (duv + u dv)$$

En appliquant la propriété de linéarité de l'opérateur \int , nous avons

$$\int d(uv) = \int v du + \int u dv$$

d'où

$$\int u dv = \int d(uv) - \int v du$$

donc

$$\int u dv = uv - \int v du \quad (1)$$

La règle [parts](#) de l'extension *Student[CalculusI]* permet de traiter la différentielle à intégrer en deux parties u et dv . La syntaxe de cette macro-commande est *Rule_{parts, u, v}* où

- la partie u est la partie dont la différentielle sera calculée par Maple
- la partie v , est celle qu'on doit obtenir (séparément) en intégrant la partie dv .

Illustrons la technique d'intégration par parties avec l'exemple suivant. Considérons l'intégrale indéfinie

$$\int x e^x dx.$$

Décomposons la différentielle à intégrer $x e^x dx$ en posant les parties suivantes:

$$\begin{array}{ll} u = x & dv = e^x dx. \\ du = & v = \end{array}$$

Le produit $u dv = x e^x dx$ correspond bien à la différentielle à intégrer. Pour compléter le premier terme du membre de droite de la formule, il nous faut déterminer v à partir de l'égalité $dv = e^x dx$. Par intégration, $\int dv = v = \int e^x dx$

La différentiation de chaque membre de l'égalité $u = x$, donne $du = dx$. Substituons ensuite ces derniers résultats dans la formule (1)

$$\int x e^x dx = x e^x - \int e^x dx$$

Reste à obtenir l'intégrale indéfinie dans le membre de droite et à globaliser les constantes d'intégration.

$$\int x e^x dx = x e^x - e^x + C$$

L'exemple 1 ci-dessous transpose en Maple le développement précédent.

Exemple 1

Calculons l'intégrale indéfinie $\int x e^x dx$.

Question d'observer le mécanisme de la simplification automatique, commençons par obtenir directement avec la macro-commande `int`, l'intégrale indéfinie $\int x e^x dx$.

Pour mieux documenter les prochains développements, créons d'abord une fonction `f` (avec l'opérateur flèche) dont la règle `f(x)` sera l'intégrande.

```
> f:=x->x*exp(x);
```

$$f := x \rightarrow x e^x \quad (4)$$

Pour que l'intégrale indéfinie puisse s'afficher avec une notation mathématique habituelle dans la zone des résultats, utilisons la forme inactive de la macro-commande `int`. De plus, donnons le nom `Problème` à l'intégrale indéfinie.

```
> Problème:=Int(f(x),x);
```

$$\text{Problème} := \int x e^x dx \quad (5)$$

Obtenons maintenant cette intégrale indéfinie avec la macro-commande `value`. Assignons le résultat au nom `primitive`.

```
> primitive:=value(Problème);
```

$$\text{primitive} := (x - 1) e^x \quad (6)$$

Rappelons ici que pour obtenir l'ensemble de toutes les primitives, il faut gérer manuellement la globalisation des constantes d'intégration à la formulation de votre réponse finale. Ajoutons donc une constante C d'intégration au résultat précédent.

```
> Rép_Maple:=Problème=primitive+C;
```

$$\text{Rép_Maple} := \int x e^x dx = (x - 1) e^x + C \quad (7)$$

Obtenons maintenant $\int x e^x dx$ avec un développement pas à pas.

Attention: Soyez attentif à la manière dont est documenté le développement pas à pas qui transpose en Maple la technique d'intégration par parties. Ce sera le modèle à suivre pour développer les intégrales indéfinies de votre TP.

Calculons $\int x e^x dx$ en intégrant par parties. Rappelons-nous l'énoncé de l'intégrale indéfinie.

```
> Problème;
```

$$\int x e^x dx \quad (8)$$

Spécifions dans la règle, la partie $u = x$. Pour spécifier la partie v , il faut intégrer dv , soit calculer $\int e^x dx$.

Ici, pas besoin de créer un sous-problème, c'est le modèle $\int e^x dx = e^x + C$.

```
> Rule[parts,x,exp(x)]((8));
Creating problem #1
```

$$\int x e^x dx = x e^x - \left(\int e^x dx \right) \quad (9)$$

L'intégrale indéfinie dans le membre de droite correspond à une formule de base. Notre but est atteint. Appliquons le modèle exp.

```
> Rule[exp]((9));
```

$$\int x e^x dx = x e^x - e^x \quad (10)$$

Factorisons e^x avant de globaliser les constantes d'intégration dans la réponse finale.

```
> Rép_finale:=Problème=factor(rhs((10)))+C;
```

$$\text{Rép_finale} := \int x e^x dx = (x - 1) e^x + C \quad (11)$$

Vérification de la réponse finale par dérivation.

```
> Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);
```

$$\frac{\partial}{\partial x} \left((x - 1) e^x + C \right) = e^x + (x - 1) e^x \quad (12)$$

```
> ``=expand(rhs((12)));
```

$$= x e^x \quad (13)$$

Cette dernière expression est bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$x e^x \quad (14)$$

Exemple 2

Calculons l'intégrale indéfinie $\int x^2 e^{2x} dx$.

Remarque: Cet exemple nous amène à répéter, dans un même calcul, deux fois la technique d'intégration par parties afin de reconnaître finalement un modèle de base dans la partie $\int v du$.

Commençons par obtenir directement avec la macro-commande int, l'intégrale indéfinie $\int x^2 e^{2x} dx$, question d'observer le mécanisme de la simplification automatique.

Pour mieux documenter les prochains développements, créons d'abord une fonction f (avec l'opérateur flèche) dont la règle f(x) sera l'intégrande.

```
> f:=x->x^2*exp(2*x);
```

$$f := x \rightarrow x^2 e^{2x} \quad (15)$$

Pour que l'intégrale indéfinie puisse s'afficher avec une notation mathématique habituelle dans la zone des résultats, utilisons la forme inactive de la macro-commande int. De plus, donnons le nom Problème à l'intégrale indéfinie.

```
> Problème:=Int(f(x),x);
```

(16)

$$\text{Problème} := \int x^2 e^{2x} dx \quad (16)$$

Obtenons maintenant cette intégrale indéfinie avec la macro-commande *value*. Assignons le résultat au nom primitive.

```
> primitive:=value(Problème);
```

$$\text{primitive} := \frac{1}{4} (2x^2 - 2x + 1) e^{2x} \quad (17)$$

Ajoutons manuellement une constante *C* d'intégration à la formulation du résultat précédent.

```
> Rép_Maple:=Problème=primitive+C;
```

$$\text{Rép_Maple} := \int x^2 e^{2x} dx = \frac{1}{4} (2x^2 - 2x + 1) e^{2x} + C \quad (18)$$

Calculons maintenant $\int x^2 e^{2x} dx$ avec un développement pas à pas. Obtenons $\int x^2 e^{2x} dx$ en intégrant par parties. Rappelons-nous l'énoncé de l'intégrale indéfinie.

```
> Problème;
```

$$\int x^2 e^{2x} dx \quad (19)$$

Posons les parties $u = x^2$ et $dv = e^{2x} dx$. Il est facile de déduire directement que $v = \frac{e^{2x}}{2}$.

```
> Rule[parts,x^2,exp(2*x)/2]((19));
```

```
Creating problem #2
```

$$\int x^2 e^{2x} dx = \frac{1}{2} x^2 e^{2x} - \left(\int x e^{2x} dx \right) \quad (20)$$

```
> Rule[parts,x,exp(2*x)/2]((20));
```

$$\int x^2 e^{2x} dx = \frac{1}{2} x^2 e^{2x} - \frac{1}{2} x e^{2x} + \int \frac{1}{2} e^{2x} dx \quad (21)$$

```
> Rule[change,u=2*x]((21));
```

```
Applying substitution x = 1/2*u, u = 2*x with dx = 1/2*du, du = 2*dx
```

$$\int x^2 e^{2x} dx = \frac{1}{2} x^2 e^{2x} - \frac{1}{2} x e^{2x} + \int \frac{1}{4} e^u du \quad (22)$$

```
> Rule[`c*`]((22));
```

$$\int x^2 e^{2x} dx = \frac{1}{2} x^2 e^{2x} - \frac{1}{2} x e^{2x} + \frac{1}{4} \int e^u du \quad (23)$$

```
> Rule[exp]((23));
```

$$\int x^2 e^{2x} dx = \frac{1}{2} x^2 e^{2x} - \frac{1}{2} x e^{2x} + \frac{1}{4} e^u \quad (24)$$

```
> Rule[revert]((24));
Reverting substitution using u = 2*x
```

$$\int x^2 e^{2x} dx = \frac{1}{2} x^2 e^{2x} - \frac{1}{2} x e^{2x} + \frac{1}{4} e^{2x} \quad (25)$$

```
> Rép_finale:=Problème=factor(rhs((25))+C;
```

$$\text{Rép_finale} := \int x^2 e^{2x} dx = \frac{1}{4} (2x^2 - 2x + 1) e^{2x} + C \quad (26)$$

Vérification de la réponse finale par dérivation.

```
> Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);
```

$$\frac{\partial}{\partial x} \left(\frac{1}{4} (2x^2 - 2x + 1) e^{2x} + C \right) = \frac{1}{4} (4x - 2) e^{2x} + \frac{1}{2} (2x^2 - 2x + 1) e^{2x} \quad (27)$$

```
> ``=normal(rhs((27)));
```

$$= x^2 e^{2x} \quad (28)$$

Cette dernière expression est bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1]
```

$$x^2 e^{2x} \quad (29)$$

Exemple 3

Calculons l'intégrale indéfinie $\int e^x \cos(x) dx$.

Remarque: Cet exemple nous amène à résoudre une équation dont l'inconnue est précisément l'intégrale indéfinie à calculer. C'est l'exemple d'un cas d'une intégration par parties cyclique.

Avec Maple directement.

```
> f:=x->exp(x)*cos(x);
```

$$f := x \rightarrow e^x \cos(x) \quad (30)$$

Posons l'intégrale indéfinie $\int e^x \cos(x) dx$ qu'il faut obtenir.

```
> Problème:=Int(f(x),x);
```

$$\text{Problème} := \int e^x \cos(x) dx \quad (31)$$

Obtenons directement cette intégrale indéfinie avec la macro-commande *value*.

```
> primitive:=value(Problème);
```

$$\text{primitive} := \frac{1}{2} e^x \cos(x) + \frac{1}{2} e^x \sin(x) \quad (32)$$

```
> Rép_Maple:=Problème=factor(primitive)+C;
```

$$\text{Rép_Maple} := \int e^x \cos(x) dx = \frac{1}{2} e^x (\cos(x) + \sin(x)) + C \quad (33)$$

Maintenant, avec un développement pas à pas. Intégrons par parties. Rappelons-nous l'énoncé de l'intégrale indéfinie.

> **Problème;**

$$\int e^x \cos(x) dx \quad (34)$$

Posons la partie $u = e^x$ et $dv = \cos(x) dx$ d'où $v = \sin(x)$.

> **Rule[parts,exp(x),sin(x)]((34));**

Creating problem #3

$$\int e^x \cos(x) dx = e^x \sin(x) - \left(\int e^x \sin(x) dx \right) \quad (35)$$

Le membre de droite $\int u dv$ contient une intégrale indéfinie de la même forme que l'intégrale indéfinie de départ. Obtenons donc cette intégrale indéfinie en intégrant de nouveau par parties. Posons encore $u = e^x$ dans et $dv = \sin(x) dx$. On déduit facilement que $v = -\cos(x)$.

> **Rule[parts,exp(x),-cos(x)]((35));**

$$\int e^x \cos(x) dx = e^x \sin(x) + e^x \cos(x) + \left(-e^x \cos(x) \right) dx \quad (36)$$

Sortons la constante multiplicatrice -1 la dernière intégrale indéfinie.

> **Rule[`c*`](36);**

$$\int e^x \cos(x) dx = e^x \sin(x) + e^x \cos(x) - \left(\int e^x \cos(x) dx \right) \quad (37)$$

Ce que nous obtenons est une équation dont l'inconnue est l'intégrale indéfinie à calculer. Résolvons donc cette équation pour $\int e^x \cos(x) dx$ (c'est l'inconnue).

> **Rule[solve]((37));**

$$\int e^x \cos(x) dx = \frac{1}{2} e^x \cos(x) + \frac{1}{2} e^x \sin(x) \quad (38)$$

Factorisons ce résultat et ajoutons immédiatement une constante d'intégration C .

> **Rép_finale:=Problème=factor(rhs((38)))+C;**

$$\text{Rép_finale} := \int e^x \cos(x) dx = \frac{1}{2} e^x (\cos(x) + \sin(x)) + C \quad (39)$$

Vérification de la réponse finale par dérivation.

> **Diff(rhs(Rép_finale),x)=diff(rhs(Rép_finale),x);**

$$\frac{\partial}{\partial x} \left(\frac{1}{2} e^x (\cos(x) + \sin(x)) + C \right) = \frac{1}{2} e^x (\cos(x) + \sin(x)) + \frac{1}{2} e^x (\cos(x) - \sin(x)) \quad (40)$$

> **``=expand(rhs((40)));**

$$= e^x \cos(x) \quad (41)$$

Cette dernière expression est bien l'intégrande de Problème. En effet,

> **Integrand(Problème)[1];**

$$e^x \cos(x) \quad (42)$$

Exemple 4

Calculons l'intégrale indéfinie $\int (\sec^3)(x) dx$.

Remarque: Cet exemple nous amène encore à une équation dont l'inconnue est l'intégrale indéfinie à

$$\int \sec(x)^3 dx = \frac{1}{2} \sec(x) \tan(x) + \frac{1}{2} \int \sec(x) dx \quad (52)$$

> **Rule[sec](52);**

$$\int \sec(x)^3 dx = \frac{1}{2} \sec(x) \tan(x) + \frac{1}{2} \ln(\sec(x) + \tan(x)) \quad (53)$$

> **Rép_finale:=Problème=rhs(53)+C;**

$$Rép_finale := \int \sec(x)^3 dx = \frac{1}{2} \sec(x) \tan(x) + \frac{1}{2} \ln(\sec(x) + \tan(x)) + C \quad (54)$$

Vérification de la réponse finale par dérivation.

> **Diff(rhs(Rép_finale),x)=diff(rhs(Problème),x);**

$$\frac{\partial}{\partial x} \left(\frac{1}{2} \sec(x) \tan(x) + \frac{1}{2} \ln(\sec(x) + \tan(x)) + C \right) = \frac{1}{2} \sec(x) \tan(x)^2 + \frac{1}{2} \sec(x) (1 + \tan(x)^2) + \frac{1}{2} \frac{\sec(x) \tan(x) + 1 + \tan(x)^2}{\sec(x) + \tan(x)} \quad (55)$$

Simplifions le membre de droite.

> **``=simplify(rhs(55),trig);**

$$= \frac{1}{\cos(x)^3} \quad (56)$$

Cette dernière expression est bien l'intégrande de Problème. En effet,

> **convert(Integrand(Problème)[1],expsincos);**

$$\frac{1}{\cos(x)^3} \quad (57)$$

REMARQUE: La simplification automatique de $\int \sec(x) dx$ a donné $\ln(\sec(x) + \tan(x))$ plutôt que d'avoir donné $\ln(|\sec(x) + \tan(x)|)$, ce qui aurait été conforme au modèle de base retenu en classe. Rappelons que c'est à vous, utilisateur de Maple, de gérer le mécanisme de la simplification automatique de Maple.

À l'aide d'un copier-coller dans une zone de texte et en éditant ce qu'il faut, on donne alors comme réponse finale:

$$Rép_finale := \int \sec(x)^3 dx = \frac{\sec(x) \tan(x)}{2} + \frac{\ln(|\sec(x) + \tan(x)|)}{2} + C$$

Ou bien, si on désire formuler la réponse finale dans une zone de résultats, on poursuit le développement depuis (51) comme suit.

> **Rule[``](50);**

$$\int \sec(x)^3 dx = \sec(x) \tan(x) - \left(\int \sec(x)^3 dx \right) + \int \sec(x) dx \quad (58)$$

> **Rule[``](50);**

$$\int \sec(x)^3 dx = \sec(x) \tan(x) - \left(\int \sec(x)^3 dx \right) + \int \sec(x) dx \quad (59)$$

```
> Rule[solve]((51));
```

$$\int \sec(x)^3 dx = \frac{1}{2} \sec(x) \tan(x) + \frac{1}{2} \int \sec(x) dx \quad (60)$$

```
> Rép:=Problème=value(rhs((52))+C;
```

$$\text{Rép} := \int \sec(x)^3 dx = \frac{1}{2} \sec(x) \tan(x) + \frac{1}{2} \ln(\sec(x) + \tan(x)) + C \quad (61)$$

```
> Rép_finale:=Problème=applyop(abs,[2,2,1],rhs((61)));
```

$$\text{Rép_finale} := \int \sec(x)^3 dx = \frac{1}{2} \sec(x) \tan(x) + \frac{1}{2} \ln(|\sec(x) + \tan(x)|) + C \quad (62)$$

C'est un peu plus compliqué que de faire tout simplement un copier-coller dans une zone de texte et d'y apporter les modifications nécessaires. Pour être en mesure de formuler la requête précédente, il faut connaître la manière dont Maple mémorise les expressions. La dernière requête a donc tout simplement pour but de montrer la possibilité de pointer interactivement vers une composante d'une expression déjà mémorisée pour y apporter ultérieurement des modifications.

Pour les intéressé(e)s, voir sur mon site Internet le document Maple intitulé *Arborescence d'un objet Maple*.

Exemple 5

Calculons l'intégrale indéfinie $\int \arcsin(x) dx$.

Remarque: Cet exemple nous amène à compléter la technique d'intégration par parties avec un changement de variables avant de reconnaître un modèle de base.

Avec Maple directement.

```
> f:=x->arcsin(x);
```

$$f := x \rightarrow \arcsin(x) \quad (63)$$

Posons l'intégrale indéfinie $\int \arcsin(x) dx$ qu'il faut obtenir.

```
> Problème:=Int(f(x),x);
```

$$\text{Problème} := \int \arcsin(x) dx \quad (64)$$

Obtenons directement cette intégrale indéfinie avec la macro-commande `value`.

```
> primitive:=value(Problème);
```

$$\text{primitive} := x \arcsin(x) + \sqrt{-x^2 + 1} \quad (65)$$

```
> Rép_Maple:=Problème=primitive+C;
```

$$\text{Rép_Maple} := \int \arcsin(x) dx = x \arcsin(x) + \sqrt{-x^2 + 1} + C \quad (66)$$

Maintenant, avec un développement pas à pas. Intégrons par parties. Rappelons-nous l'énoncé de l'intégrale indéfinie.

```
> Problème;
```

$$\int \arcsin(x) dx \quad (67)$$

Posons la partie $u = \arcsin(x)$ et $dv = dx$.

```
> Rule[parts,arcsin(x),x]((67));
```

Creating problem #5

$$\int \arcsin(x) \, dx = x \arcsin(x) - \left(\int \frac{x}{\sqrt{-x^2+1}} \, dx \right) \quad (68)$$

Le membre de droite contient une intégrale indéfinie qui doit être évalué avec un changement de variables.

Posons le changement de variables $u = 1 - x^2$.

> Rule[change,u=1-x^2]((68));

Applying substitution $x = (-u+1)^{(1/2)}$, $u = -x^2+1$ with $dx = -1/2/(-u+1)^{(1/2)}*du$, $du = -2*x*dx$

$$\int \arcsin(x) \, dx = x \arcsin(x) - \left(\int \left(-\frac{1}{2\sqrt{u}} \right) du \right) \quad (69)$$

Sortons la constante multiplicatrice $-\frac{1}{2}$.

> Rule[`c*`](69);

$$\int \arcsin(x) \, dx = x \arcsin(x) + \frac{1}{2} \int \frac{1}{\sqrt{u}} \, du \quad (70)$$

> Rule[power]((70));

$$\int \arcsin(x) \, dx = x \arcsin(x) + \sqrt{u} \quad (71)$$

Considérons à rebours le changement de variables.

> Rule[revert]((71))+(0=C);

Reverting substitution using $u = -x^2+1$

$$\int \arcsin(x) \, dx = x \arcsin(x) + \sqrt{-x^2+1} + C \quad (72)$$

Vérification de la réponse finale par dérivation.

> Diff(rhs((72)),x)=diff(rhs((72)),x);

$$\frac{\partial}{\partial x} \left(x \arcsin(x) + \sqrt{-x^2+1} + C \right) = \arcsin(x) \quad (73)$$

Cette dernière expression est bien l'intégrande de Problème. En effet,

> Integrand(Problème)[1];

$$\arcsin(x) \quad (74)$$

Exemple 6

Calculons l'intégrale indéfinie $\int (\arccos^2)(x) \, dx$.

Avec Maple directement.

> f:=x->(arccos^2)(x);

$$f := x \rightarrow (\arccos^2)(x) \quad (75)$$

Posons l'intégrale indéfinie $\int (\arccos^2)(x) \, dx$ qu'il faut obtenir.

> Problème:=Int(f(x),x);

(76)

$$\text{Problème} := \int \arccos(x)^2 dx \quad (76)$$

Obtenons directement cette intégrale indéfinie avec la macro-commande `value`.

```
> primitive:=value(Problème);
```

$$\text{primitive} := \arccos(x)^2 x - 2x - 2 \arccos(x) \sqrt{-x^2 + 1} \quad (77)$$

```
> Rép_Maple:=Problème=primitive+C;
```

$$\text{Rép_Maple} := \int \arccos(x)^2 dx = \arccos(x)^2 x - 2x - 2 \arccos(x) \sqrt{-x^2 + 1} + C \quad (78)$$

Calculons cette intégrale indéfinie avec la technique d'intégration par parties. Rappelons-nous l'énoncé de l'intégrale indéfinie qu'on cherche à obtenir.

```
> Problème;
```

$$\int \arccos(x)^2 dx \quad (79)$$

Posons les parties $u = \arccos(x)$ et $dv = \arccos(x) dx$. Nous allons créer un sous-problème afin d'obtenir v .

```
> Sous_Problème_1:=Int(arccos(x),x);
```

$$\text{Sous_Problème}_1 := \int \arccos(x) dx \quad (80)$$

Posons les parties $u = \arccos(x)$ et $dv = dx$.

```
> Rule[parts,arccos(x),x]((80));
```

Creating problem #6

$$\int \arccos(x) dx = \arccos(x) x - \left(\int \left(-\frac{x}{\sqrt{-x^2 + 1}} \right) dx \right) \quad (81)$$

```
> Rule[`c*`](81);
```

$$\int \arccos(x) dx = \arccos(x) x + \int \frac{x}{\sqrt{-x^2 + 1}} dx \quad (82)$$

```
> Rule[change,u=1-x^2]((82));
```

Applying substitution $x = (-u+1)^{(1/2)}$, $u = -x^2+1$ with $dx = -1/2/(-u+1)^{(1/2)}*du$, $du = -2*x*dx$

$$\int \arccos(x) dx = \arccos(x) x + \int \left(-\frac{1}{2\sqrt{u}} \right) du \quad (83)$$

```
> Rule[`c*`](83);
```

$$\int \arccos(x) dx = \arccos(x) x - \frac{1}{2} \int \frac{1}{\sqrt{u}} du \quad (84)$$

```
> Rule[power]((84));
```

$$\int \arccos(x) dx = \arccos(x) x - \sqrt{u} \quad (85)$$

> Rule[revert]((85));

Reverting substitution using $u = -x^2+1$

$$\int \arccos(x) dx = \arccos(x) x - \sqrt{-x^2 + 1} \quad (86)$$

> Problème;

$$\int \arccos(x)^2 dx \quad (87)$$

Posons les parties $u = \arccos(x)$ et $dv = \arccos(x) dx$.

> Rule[parts,arccos(x),arccos(x)*x - sqrt(-x^2 + 1)](Problème);

Creating problem #7

$$\int \arccos(x)^2 dx = \arccos(x) \left(\arccos(x) x - \sqrt{-x^2 + 1} \right) - \left(\int \left(-\frac{\arccos(x) x - \sqrt{-x^2 + 1}}{\sqrt{-x^2 + 1}} \right) dx \right) \quad (88)$$

> expand((88));

$$\int \arccos(x)^2 dx = \arccos(x)^2 x - \arccos(x) \sqrt{-x^2 + 1} + \int \left(\frac{\arccos(x) x}{\sqrt{-x^2 + 1}} - 1 \right) dx \quad (89)$$

> Rule[``]((89));

$$\int \arccos(x)^2 dx = \arccos(x)^2 x - \arccos(x) \sqrt{-x^2 + 1} + \int \frac{\arccos(x) x}{\sqrt{-x^2 + 1}} dx + \int (-1) dx \quad (90)$$

> Rule[constant]((90));

$$\int \arccos(x)^2 dx = \arccos(x)^2 x - \arccos(x) \sqrt{-x^2 + 1} + \int \frac{\arccos(x) x}{\sqrt{-x^2 + 1}} dx - x \quad (91)$$

Intégrons $\int \frac{x \arccos(x)}{\sqrt{1-x^2}} dx$ par parties en posant la partie $u = \arccos(x)$ et $dv = \frac{x}{\sqrt{1-x^2}} dx$. Obtenons v pour être en mesure d'utiliser la règle *parts*.

```
> Rule[change,u=1-x^2](Int(x/sqrt(-x^2 + 1),x));
```

```
Creating problem #8
```

```
Applying substitution x = (-u+1)^(1/2), u = -x^2+1 with dx = -1/2/(-u+1)^(1/2)*du, du = -2*x*dx
```

$$\int \frac{x}{\sqrt{-x^2+1}} dx = \int \left(-\frac{1}{2\sqrt{u}} \right) du \quad (92)$$

```
> Rule[`c*`](92);
```

$$\int \frac{x}{\sqrt{-x^2+1}} dx = -\frac{1}{2} \int \frac{1}{\sqrt{u}} du \quad (93)$$

```
> Rule[power](93);
```

$$\int \frac{x}{\sqrt{-x^2+1}} dx = -\sqrt{u} \quad (94)$$

```
> Rule[revert](94);
```

```
Reverting substitution using u = -x^2+1
```

$$\int \frac{x}{\sqrt{-x^2+1}} dx = -\sqrt{-x^2+1} \quad (95)$$

Intégrons maintenant par parties

```
> Rule[parts,arccos(x),-sqrt(-x^2 + 1)](91);
```

$$\int \arccos(x)^2 dx = \arccos(x)^2 x - 2 \arccos(x) \sqrt{-x^2+1} - \left(\int 1 dx \right) - x \quad (96)$$

```
> Rule[constant](96);
```

$$\int \arccos(x)^2 dx = \arccos(x)^2 x - 2x - 2 \arccos(x) \sqrt{-x^2+1} \quad (97)$$

```
> Rép_finale:=Problème=rhs(97)+C;
```

$$\text{Rép_finale} := \int \arccos(x)^2 dx = \arccos(x)^2 x - 2x - 2 \arccos(x) \sqrt{-x^2+1} + C \quad (98)$$

Vérification de la réponse finale par dérivation.

```
> Diff(rhs(98),x)=diff(rhs(98),x);
```

$$\frac{\partial}{\partial x} \left(\arccos(x)^2 x - 2x - 2 \arccos(x) \sqrt{-x^2+1} + C \right) = \arccos(x)^2 \quad (99)$$

Cette dernière expression est bien l'intégrande de Problème. En effet,

```
> Integrand(Problème)[1];
```

$$\arccos(x)^2 \quad (100)$$

